> # **exercice 1** : etude de la suite récurrente x[n+1]=x[n]+n/x[n]#

> # On a pris x[1]=2#

```
> x[1]:=2: for k from 2 to 30 do x[k]:=evalf(x[k-1]+(k-1)/x[k-1]) od:
```

> # on représente la suite x[n]#

```
> plot([seq([n,x[n]],n=1..30)]);
```



> #On peut penser qu'un équivalent de cette suite est n On représente la suite x[n]-n puis la suite de terme n(x[n]-n)#

```
> plot([seq([i,x[i]-i],i=1..30)]);
```



```
> with(plots): plot([seq([j,j*(x[j]-j)],j=1..30)],view=[0..30,0..1]);
```



> #Conclusion: on peut chercher un développement asymptotique de la forme x[n]=n+c/n où c est une constante qui dépend de x[1]#

>

> #Exercice 2 calcul d'une intégrale#

> f:=p->1/(sqrt(1+x^p)+sqrt(1-x^p));

$$f := p \to \frac{1}{\sqrt{1+x^p}+\sqrt{1-x^p}}$$

> #Remarquer que le paramétre p est considéré comme variable, ainsi f(p) est une expression de x#

> r:=int(f(1),x=0..1);

$$r := \sqrt{2} + \frac{1}{2}\ln(\sqrt{2}-1) - \frac{1}{2}\ln(1+\sqrt{2})$$

> s:=int(f(2),x=0..1);

$$s := -\frac{1}{2}\sqrt{2} - \frac{1}{2}\ln(\sqrt{2}-1) + \frac{1}{4}\pi$$

> #Calcul d'une valeur approchée de l'intégrale de f(p) pour p variant de 1 à 10#

> for k from 1 to 10 do evalf(int(f(k), x=0..1)) od;

$$0.5328399746$$

$$0.5189781765$$

$$0.5133693004 - 0.00001924579801\, I$$

$$0.5103197779 + 8.333333333\ 10^{-11}\, I$$

$$0.5084051998$$

$$0.5070902361$$

$$0.5061312218$$

$$0.5054008102$$

$$0.5048259458$$

$$0.5043617118$$

> # On peut penser que la suite tend vers 1/2 La valeur de l'intégrale de f(3) On peut demander une valeur exacte#

> int(f(3),x=0..1);

$$\begin{aligned}
&\frac{1}{8\sqrt{3+I\sqrt{3}}}\Bigg(-9\sqrt{-\frac{2}{-3+I\sqrt{3}}}\sqrt{\frac{1+I\sqrt{3}}{3+I\sqrt{3}}}\sqrt{\frac{-1+I\sqrt{3}}{-3+I\sqrt{3}}}\,\mathrm{EllipticF}\left(\sqrt{-\frac{2}{-3+I\sqrt{3}}},\sqrt{-\frac{-3+I\sqrt{3}}{3+I\sqrt{3}}}\right)\sqrt{3+I\sqrt{3}}\\
&+3I\sqrt{-\frac{2}{-3+I\sqrt{3}}}\sqrt{\frac{1+I\sqrt{3}}{3+I\sqrt{3}}}\sqrt{\frac{-1+I\sqrt{3}}{-3+I\sqrt{3}}}\,\mathrm{EllipticF}\left(\sqrt{-\frac{2}{-3+I\sqrt{3}}},\sqrt{-\frac{-3+I\sqrt{3}}{3+I\sqrt{3}}}\right)\sqrt{3}\sqrt{3+I\sqrt{3}}\\
&-3I\sqrt{-I(1+I\sqrt{3})}\sqrt{2}\sqrt{-I(-1+I\sqrt{3})}\,\mathrm{EllipticF}\left(\frac{1}{6}\sqrt{2}\sqrt{-I\sqrt{3}+3}\sqrt{3},\sqrt{2}\,3^{(1/4)}\sqrt{\frac{I}{3+I\sqrt{3}}}\right)-2\sqrt{2}\sqrt{3+I\sqrt{3}}\\
&+I\sqrt{3}\sqrt{-I(3+I\sqrt{3})}\sqrt{2}\sqrt{-I(-3+I\sqrt{3})}\,\mathrm{EllipticF}\left(\frac{1}{6}\sqrt{2}\sqrt{-3I\sqrt{3}+3}\sqrt{3},\sqrt{2}\,3^{(1/4)}\sqrt{\frac{I}{3+I\sqrt{3}}}\right)\\
&+9\sqrt{-\frac{1}{-3+I\sqrt{3}}}\sqrt{\frac{-1+I\sqrt{3}}{3+I\sqrt{3}}}\sqrt{\frac{1+I\sqrt{3}}{-3+I\sqrt{3}}}\sqrt{2}\,\mathrm{EllipticF}\left(\sqrt{2}\sqrt{-\frac{2}{-3+I\sqrt{3}}},\sqrt{-\frac{-3+I\sqrt{3}}{3+I\sqrt{3}}}\right)\sqrt{3+I\sqrt{3}}\\
&-3I\sqrt{-\frac{1}{-3+I\sqrt{3}}}\sqrt{\frac{-1+I\sqrt{3}}{3+I\sqrt{3}}}\sqrt{\frac{1+I\sqrt{3}}{-3+I\sqrt{3}}}\sqrt{2}\,\mathrm{EllipticF}\left(\sqrt{2}\sqrt{-\frac{2}{-3+I\sqrt{3}}},\sqrt{-\frac{-3+I\sqrt{3}}{3+I\sqrt{3}}}\right)\sqrt{3}\sqrt{3+I\sqrt{3}}\Bigg)
\end{aligned}$$

> evalf(Int(f(3),x=0..1));

$$0.5133664899$$

> #la bonne syntaxe est donc evalf(Int..) pour que Maple utilise les methodes de calcul approché d'une intégrale, sinon il calcule la valeur exacte, ici fort complexe et il en cherche une valeur approchée, d'où les erreurs de calcul#

>

> #Exercice 3 Etude de la fonction f(x) est la somme de la série de terme exp(-x^2n^2)#

> restart:f:=x->sum(exp(-x^2*n^2), n=0..+infinity);

$$f := x \to \sum_{n=0}^{\infty} \mathrm{e}^{(-x^2 n^2)}$$

> f(x);

$$\sum_{n=0}^{\infty} \mathrm{e}^{(-x^2 n^2)}$$

> plot(f(x),x=0..1);



> plot(x*f(x), x=0..1,y=0..3);



> evalf(0.1*f(0.1));

$$0.1\left(\sum_{n=0}^{\infty} e^{(-0.01000000000\, n^2)}\right)$$

> evalf(0.1*sum(exp(-0.01*k^2), k=0..30));

$$0.9362129047$$

> evalf(sqrt(Pi)/2);

$$0.8862269255$$

> # Conclusion, il est difficile de calculer des valeurs approchées de f(x) quand x est petit, d'où la prudence de Maple qui ne donne pas evalf(0.1*f(0.1))#

>

> #**Exercice 4** Calcul de sommes#

> S1:=sum(1/((4*n+1)*(4*n+3)), n=0..+infinity);

$$S1 := \frac{1}{8}\pi$$

> S2:=sum((-1)^n/((4*n+1)*(4*n+3)),n=0..+infinity);

$$S2 := \frac{1}{4}\sqrt{2}\ln(\sqrt{2}+1)$$

> #Cette version de Maple donne la valeur de S2 On peut montrer que S2 est l'intégrale suivante#

> r:=int((1-t^2)/(2*(1+t^4)),t=0..1);

$$r := \frac{1}{8}\sqrt{2}\ln(\sqrt{2}+2) - \frac{1}{8}\sqrt{2}\ln(-\sqrt{2}+2)$$

> simplify(r);

$$-\frac{1}{8}\sqrt{2}\,(-\ln(\sqrt{2}+2)+\ln(-\sqrt{2}+2))$$

> simplify(ln((2+sqrt(2))/(2-sqrt(2))));

$$\ln(3+2\sqrt{2})$$

> expand((sqrt(2)+1)^2);

$$3+2\sqrt{2}$$

> # On retrouve le même résultat#

>

> #**Exercice 5** Résolution d'un système différentiel#

> restart: eq1:=diff(f(x),x)=3*f(x)-2*g(x); eq2:=diff(g(x),x)=2*f(x)+3*g(x);

$$eq1 := \frac{d}{dx}f(x) = 3f(x) - 2g(x)$$

$$eq2 := \frac{d}{dx}g(x) = 2f(x) + 3g(x)$$

> dsolve({eq1,eq2,f(0)=a,g(0)=b},{f(x),g(x)});

$$\{g(x) = -e^{(3x)}(-b\cos(2x) - a\sin(2x)), f(x) = e^{(3x)}(-b\sin(2x) + a\cos(2x))\}$$

> # On peut trouver la forme des solutions en cherchant les valeurs propres de la matrice des coefficients#

> with(linalg):

Warning, the protected names norm and trace have been redefined and unprotected

```
> A:=matrix(2,2,[3,-2,2,3]);
```

$$A := \begin{bmatrix} 3 & -2 \\ 2 & 3 \end{bmatrix}$$

```
> eigenvals(A);
```

$$3+2I, 3-2I$$

> #**Exercice 6** Etude de la suite récurrente u[n+2]=(n+1)u[n+1]-(n+2)u[n] avec u[0]=u[1]=-1#

```
> restart: u[0]:=-1; u[1]:=-1;
```

$$u_0 := -1$$

$$u_1 := -1$$

```
> for k from 2 to 10 do u[k]:=(k-1)*u[k-1]-k*u[k-2] od;
```

$$u_2 := 1$$

$$u_3 := 5$$

$$u_4 := 11$$

$$u_5 := 19$$

$$u_6 := 29$$

$$u_7 := 41$$

$$u_8 := 55$$

$$u_9 := 71$$

$$u_{10} := 89$$

> # on pose f(x), la somme de la série de terme u[n]x^n et on trouve une équation différentielle vérifiée par f#

```
> eq:=(1-x)*x^2*diff(y(x),x)-(1+2*x^2)*y(x)=1+x;
```

$$eq := (1-x)\,x^2\left(\frac{d}{dx}y(x)\right) - (1+2\,x^2)\,y(x) = 1+x$$

```
> r:=dsolve({eq},{y(x)});
```

$$r := \left\{ y(x) = \frac{1-2\,x-x^2+e^{\left(-\frac{1}{x}\right)}\,\_C1\,x}{(-1+x)^3} \right\}$$

> # cette équation admet une unique solution développable en série entière pour la valeur de la constante _C1=0#

```
> f:=unapply(subs(_C1=0,rhs(op(r))),x);
```

$$f := x \to \frac{1-2\,x-x^2}{(-1+x)^3}$$

> # Décomposons en éléments simple f #

> g:=x->f(x)-a/(1-x)-b/(1-x)^2-c/(1-x)^3;

$$g := x \to f(x) - \frac{a}{1-x} - \frac{b}{(1-x)^2} - \frac{c}{(1-x)^3}$$

> s:=solve({g(0)=0, g(2)=0,g(3)=0},{a,b,c});

$$s := \{a = 1, c = 2, b = -4\}$$

> assign(s);

> # On connait le développement de 1/(1-x)^k#

> expand(a+b*(n+1)+c*(n+2)*(n+1)/2);

$$-1 - n + n^2$$

> # La valeur de u[n] est n^2-n-1#

>

> **#Exercice 7** Racines d'un polynôme#

> restart: P:= 6*x^6+4*x^5+3*x^4+10*x^3+3*x^2+4*x+6;

$$P := 6x^6 + 4x^5 + 3x^4 + 10x^3 + 3x^2 + 4x + 6$$

> solve(P=0,x);

$-1, -1, \mathrm{RootOf}(6\_Z^4 - 8\_Z^3 + 13\_Z^2 - 8\_Z + 6, index = 1), \mathrm{RootOf}(6\_Z^4 - 8\_Z^3 + 13\_Z^2 - 8\_Z + 6, index = 2), \mathrm{RootOf}(6\_Z^4 - 8\_Z^3 + 13\_Z^2 - 8\_Z + 6, index = 3), \mathrm{RootOf}(6\_Z^4 - 8\_Z^3 + 13\_Z^2 - 8\_Z + 6, index = 4)$

> # On peut donc écrire P sous forme (x-1)^2Q(x)avec#

> Q:= 6*x^4-8*x^3+13*x^2-8*x+6;

$$Q := 6x^4 - 8x^3 + 13x^2 - 8x + 6$$

> # Ce polynôme est symétrique, on l'exprime en fonction de x+1/x#

> R:=expand(Q/x^2);

$$R := 6x^2 - 8x + 13 - \frac{8}{x} + \frac{6}{x^2}$$

> y:=x+1/x; expand(y^2);

$$y := x + \frac{1}{x}$$

$$x^2 + 2 + \frac{1}{x^2}$$

> # d'où le polynôme#

> y:='y';S:=6*y^2-8*y+1;

$$y := y$$

$$S := 6y^2 - 8y + 1$$

> # on libère la variable y, sinon S serait fonction de x#

> u:=solve(S,y);

$$u := \frac{2}{3} + \frac{1}{6}\sqrt{10}, \frac{2}{3} - \frac{1}{6}\sqrt{10}$$

> v:=solve(x+1/x=u[1],x); w:=solve(x+1/x=u[2],x);

$$v := \frac{1}{3} + \frac{1}{12}\sqrt{10} + \frac{1}{12}\sqrt{-118+8\sqrt{10}}, \frac{1}{3} + \frac{1}{12}\sqrt{10} - \frac{1}{12}\sqrt{-118+8\sqrt{10}}$$

$$w := \frac{1}{3} - \frac{1}{12}\sqrt{10} + \frac{1}{12}\sqrt{-118-8\sqrt{10}}, \frac{1}{3} - \frac{1}{12}\sqrt{10} - \frac{1}{12}\sqrt{-118-8\sqrt{10}}$$

> #d'où les racines de P -1 à l'ordre 2 et#

> evalc(v[1]); evalc(v[2]); evalc(w[1]); evalc(w[2]);

$$\frac{1}{3} + \frac{1}{12}\sqrt{10} + \frac{1}{12} I \sqrt{118-8\sqrt{10}}$$

$$\frac{1}{3} + \frac{1}{12}\sqrt{10} - \frac{1}{12} I \sqrt{118-8\sqrt{10}}$$

$$\frac{1}{3} - \frac{1}{12}\sqrt{10} + \frac{1}{12} I \sqrt{118+8\sqrt{10}}$$

$$\frac{1}{3} - \frac{1}{12}\sqrt{10} - \frac{1}{12} I \sqrt{118+8\sqrt{10}}$$

> **#Exercice 8** Calcul d'intégrales#

> restart: f:=(n,a)-> x^n*(1-x)^n/((1-(1-a)*x)^(n+1));

$$f := (n, a) \to \frac{x^n (1-x)^n}{(1-(1-a)\,x)^{(n+1)}}$$

> # f est une expression, les paramètres sont a et n Par exemple pour a=1/5, 1/4, 1/3, 1/2#

> for k from 1 to 3 do int(f(k,1/5), x=0..1) od;

$$\frac{75}{32}\ln(5) - \frac{25}{8}$$

$$\frac{2875}{512}\ln(5) - \frac{1125}{128}$$

$$\frac{61875}{4096}\ln(5) - \frac{74375}{3072}$$

> for k from 1 to 3 do int(f(k,1/4), x=0..1) od;

$$\frac{160}{27}\ln(2) - \frac{32}{9}$$

$$\frac{1408}{81}\ln(2) - \frac{320}{27}$$

$$\frac{125440}{2187}\ln(2) - \frac{28928}{729}$$

> for k from 1 to 3 do int(f(k,1/3), x=0..1) od;

$$\frac{9}{2}\ln(3) - \frac{9}{2}$$

$$\frac{297}{16}\ln(3) - \frac{81}{4}$$

$$\frac{1377}{16}\ln(3) - \frac{189}{2}$$

> for k from 1 to 3 do int(f(k,1/2), x=0..1) od;

$$12\ln(2) - 8$$

$$104\ln(2) - 72$$

$$1008\ln(2) - \frac{2096}{3}$$

> # On peut penser que cette intégrale est de la forme rln(a)+s avec r et s rationnels#